// File: /contracts/bridge/LiquidCryptoBridge_v2.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.8.9 <0.9.0;

import '@openzeppelin/contracts/token/ERC20/ERC20.sol';
import "@openzeppelin/contracts/access/Ownable.sol";

import "./../interface/IWETH.sol";
import "./../interface/IUniswapRouterETH.sol";

contract LiquidCryptoBridge_v2 is ERC20, Ownable {
  uint256 public swapFee = 500;
  uint256 public platformFee = 30000;
  uint256 public feeDecimals = 100000;
  address public unirouter;
  address public native;
  address public stable;
  address[] public nativeToStable;
  address public treasury;

  uint256 private totalEarned = 0;
  uint256 private totalReward = 0;
  uint256 private lastTotalTs;
  uint256 private totalUserShare = 0;
  uint256 private totalUserClaimed = 0;

  uint256 public swapIndex = 0;

  struct SwapVoucher {
    uint256 iAmount;
    uint256 stableAmount;
    uint256 outChain;
    address toAccount;
    address refundAccount;
  }

  struct StakeInfo {
    uint256 amount;
    uint256 startedAt;
    uint256 debtReward;
  }

  mapping (uint256 => SwapVoucher) public voucherLists;
  mapping (address => bool) public managers;
  mapping (address => StakeInfo) public userInfo;

  event Stake(address account, uint256 amount, uint256 totalStake, uint256 totalShare);
  event Unstake(address account, uint256 amount, uint256 totalStake, uint256 totalShare);
  event Claimed(address account, uint256 amount);
  event Swap(uint256 index, address from, address to, uint256 outChain, uint256 inAmout, uint256 stableAmount);
  event Redeem(address to, uint256 stableAmount, uint256 outAmout);
  event Refund(uint256 index, address to, uint256 amount);
  event APY(uint256 totalStake, uint256 totalReward);

  constructor(
    address _unirouter,
    address[] memory _nativeToStable,
    address _treasury
  )
    ERC20("LCBridgeLP_v2", "BPv2")
  {
    unirouter = _unirouter;
    
    native = _nativeToStable[0];
    stable = _nativeToStable[_nativeToStable.length - 1];
    nativeToStable = _nativeToStable;

    treasury = _treasury;

    managers[msg.sender] = true;
    lastTotalTs = block.timestamp;
  }

  modifier onlyManager() {
    require(managers[msg.sender], "LC Bridge v2: !manager");
    _;
  }

  function swap(address _to, address _refund, uint256 _outChainID) public payable returns(uint256) {
    uint256 amount = msg.value;
    address account = msg.sender;

    amount = _cutFee(amount);
    uint256[] memory amounts = IUniswapRouterETH(unirouter).getAmountsOut(amount, nativeToStable);

    swapIndex ++;
    uint256 stableAmount = amounts[amounts.length-1];
    voucherLists[swapIndex] = SwapVoucher(amount, stableAmount, _outChainID, _to, _refund);
    emit Swap(swapIndex, account, _to, _outChainID, msg.value, stableAmount);
    emit APY(totalSupply(), totalEarned);
    return stableAmount;
  }

  function redeem(uint256 _amount, address _to, uint256 _fee, bool wrapped) public onlyManager returns(uint256){
    uint256[] memory amounts = IUniswapRouterETH(unirouter).getAmountsIn(_amount, nativeToStable);
    uint256 amount = amounts[0];
    require(amount <= address(this).balance, "LC Bridge v2: Few redeem liquidity");
    require(amount >= _fee, "LC Bridge v2: Few redeem liquidity");
    amount -= _fee;

    if (amount > 0) {
      if (wrapped) {
        IWETH(native).deposit{value: amount}();
        IWETH(native).transfer(_to, amount);
      }
      else {
        (bool success, ) = payable(_to).call{value: amount}("");
        require(success, "LC Bridge v2: Failed redeem");
      }
    }
    if (_fee > 0) {
      (bool success, ) = payable(msg.sender).call{value: _fee}("");
      require(success, "LC Bridge v2: Failed refund manager fee");
    }
    emit Redeem(_to, _amount, amount);
    return amount;
  }

  function refund(uint256 _index, uint256 _fee) public onlyManager {
    uint256 amount = voucherLists[_index].iAmount;
    (bool success, ) = payable(voucherLists[_index].refundAccount).call{value: amount - _fee}("");
    require(success, "LC Bridge v2: Failed refund");
    if (_fee > 0) {
      (success, ) = payable(msg.sender).call{value: _fee}("");
      require(success, "LC Bridge v2: Failed refund manager fee");
    }
  }

  function getAmountsIn(uint256 _amount) public view returns(uint256 coin) {
    uint256[] memory amounts = IUniswapRouterETH(unirouter).getAmountsIn(_amount, nativeToStable);
    coin = amounts[0];
  }

  function getAmountsOut(uint256 _amount) public view returns(uint256 stableAmount) {
    uint256[] memory amounts = IUniswapRouterETH(unirouter).getAmountsOut(_amount, nativeToStable);
    stableAmount = amounts[amounts.length-1];
  }

  function stake() public payable {
    uint256 amount = msg.value;
    address account = msg.sender;
    uint256 _now = block.timestamp;
    
    totalUserShare += totalSupply() * (_now - lastTotalTs);
    lastTotalTs = _now;
    StakeInfo storage tmpInfo = userInfo[account];
    tmpInfo.debtReward += tmpInfo.amount * (_now - tmpInfo.startedAt);
    tmpInfo.amount += amount;
    tmpInfo.startedAt = _now;

    _mint(msg.sender, amount);

    emit Stake(account, amount, totalSupply(), totalUserShare);
    emit APY(totalSupply(), totalEarned);
  }

  function unstake(uint256 _amount) public {
    require(_amount <= balanceOf(msg.sender), "LC Bridge v2: Exceed amount");
    require(_amount <= address(this).balance, "LC Bridge v2: Few unstake liquidity");

    uint256 _now = block.timestamp;
    address account = msg.sender;
    uint256 userReward = getReward(msg.sender);
    if (userReward > 0) {
      claimReward(msg.sender, userReward);
    }
    StakeInfo storage tmpInfo = userInfo[account];
    tmpInfo.amount -= _amount;
    totalUserShare += totalSupply() * (_now - lastTotalTs);
    lastTotalTs = _now;

    if (_amount > address(this).balance) {
      _amount = address(this).balance;
    }
    _burn(account, _amount);

    (bool success, ) = msg.sender.call{value: _amount}("");
    require(success, "LC Bridge v2: Failed to unstake");

    emit Unstake(account, _amount, totalSupply(), totalUserShare);
    emit APY(totalSupply(), totalEarned);
  }

  function getReward(address _account) public view returns(uint256) {
    uint256 _now = block.timestamp;
    uint256 tmpTotalShare = totalUserShare + totalSupply() * (_now - lastTotalTs);
    if (tmpTotalShare <= totalUserClaimed) return 0;
    tmpTotalShare -= totalUserClaimed;
    uint256 tmpUserTotal = userInfo[_account].debtReward + userInfo[_account].amount * (_now - userInfo[_account].startedAt);
    return totalReward * tmpUserTotal / tmpTotalShare;
  }

  function claimReward(address _account, uint256 _amount) public {
    uint256 userReward = getReward(_account);
    if (userReward < _amount) {
      _amount = userReward;
    }
    if (_amount > address(this).balance) {
      _amount = address(this).balance;
    }

    uint256 _now = block.timestamp;
    StakeInfo storage tmpInfo = userInfo[_account];
    tmpInfo.debtReward = (tmpInfo.debtReward + tmpInfo.amount * (_now - tmpInfo.startedAt)) * (userReward - _amount) / userReward;
    tmpInfo.startedAt = _now;
    uint256 tmpTotalShare = totalUserShare + totalSupply() * (_now - lastTotalTs);
    if (tmpTotalShare < totalUserClaimed) {
      totalUserClaimed = tmpTotalShare;
    }
    else {
     tmpTotalShare -= totalUserClaimed;
     totalUserClaimed += (tmpTotalShare * _amount / totalReward);
    }
    totalReward -= _amount;

    if (_amount > 0) {
      (bool success, ) = payable(_account).call{value: _amount}("");
      require(success, "LC Bridge v2: Failed to claim reward");
    }

    emit Claimed(_account, _amount);
  }
  
  function setManager(address account, bool access) public onlyOwner {
    managers[account] = access;
  }

  function setSwapFee(uint256 _swapFee) public onlyManager {
    swapFee = _swapFee;
  }

  function setPlatformFee(uint256 _platformFee) public onlyManager {
    platformFee = _platformFee;
  }

  function setUnirouterInfo(address _unirouter, address[] memory _nativeToStable) public onlyManager {
    unirouter = _unirouter;
    native = _nativeToStable[0];
    stable = _nativeToStable[_nativeToStable.length - 1];
    nativeToStable = _nativeToStable;
  }

  function setTreasury(address _treasury) public onlyManager {
    treasury = _treasury;
  }

  function deposit() public payable onlyManager {
    if (totalSupply() > address(this).balance) {
      uint256 needAmount = totalSupply() - address(this).balance;
      if (msg.value > needAmount) {
        uint256 overflw = msg.value - needAmount;
        (bool success, ) = msg.sender.call{value: overflw}("");
        require(success, "LC Bridge v2: Failed depoist overflow");
      }
    }
  }

  function withdraw() public onlyManager {
    if (totalSupply() < address(this).balance) {
      uint256 availableAmount = address(this).balance - totalSupply();
      (bool success1, ) = msg.sender.call{value: availableAmount}("");
      require(success1, "LC Bridge v2: Failed revoke");
    }
  }

  function _cutFee(uint256 _amount) internal returns(uint256) {
    if (_amount > 0) {
      uint256 fee = _amount * swapFee / feeDecimals;
      uint256 treasuryFee = fee * platformFee / feeDecimals;
      if (treasuryFee > 0) {
        (bool success, ) = payable(treasury).call{value: treasuryFee}("");
        require(success, "LC Bridge v2: Failed cut fee");
      }
      totalEarned += (fee - treasuryFee);
      totalReward += (fee - treasuryFee);
      return _amount - fee;
    }
    return 0;
  }
}

// File: /contracts/interface/IWETH.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.8.9 <0.9.0;

import "@openzeppelin/contracts/interfaces/IERC20.sol";

interface IWETH is IERC20 {
    function deposit() external payable;
    function withdraw(uint256 wad) external;
}
// File: /contracts/interface/IUniswapRouterETH.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.9.0;

interface IUniswapRouterETH {
    function swapExactETHForTokens(uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        payable
        returns (uint[] memory amounts);
    
    function swapExactTokensForETH(uint amountIn, uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        returns (uint[] memory amounts);

    function swapExactTokensForTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);

    function swapTokensForExactTokens(
        uint amountOut,
        uint amountInMax,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);

    function getAmountsOut(uint amountIn, address[] calldata path) external view returns (uint[] memory amounts);
    function getAmountsIn(uint amountOut, address[] calldata path) external view returns (uint[] memory amounts);
}
// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.zeppelin.solutions/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * The default value of {decimals} is 18. To select a different value for
     * {decimals} you should overload it.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless this function is
     * overridden;
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `from` to `to`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
        }
        _balances[to] += amount;

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        _balances[account] += amount;
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
        }
        _totalSupply -= amount;

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}
}

// File: @openzeppelin/contracts/interfaces/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (interfaces/IERC20.sol)

pragma solidity ^0.8.0;

import "../token/ERC20/IERC20.sol";

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}